MICRO™

The Magazine of the APPLE, KIM, PET

and Other 6502 Systems



SYMPHONY IN STEREO

NO 13 June 1979

SPECIAL $1.50
$2.00

KIM — The Tunesmith

A number of programs have been offered which permit you to play music on your micro. The program presented here also permits you to compose music on your KIM, as well as save it and play it back.

Anthony T. Scarpelli
RR 1, Box 426
N. Windham, ME 04062

Have you ever wanted to compose music, but knew nothing about how to go about doing it? Do you lack a musical instrument and have a tune going through your head and don't know what should go after the first few notes? Well here is a program for a basic KIM-1 that will help you compose a tune, and you don't even have to know how to read or write music.

I have really never learned how to play a musical instrument, and I never have time to practice. Yet every once in a while I want to try out a few notes going on in my head, or I just want to see how a couple of notes sound together, to see if they have any effect on me. So what I did was to develop a program that uses a basic KIM-1 and the speaker circuit shown on page 57 of the KIM-1 User Manual that plays a tune I compose one note at a time. I use the keypad as data entry to place into the program notes of two octaves, including sharp notes, with four possible lengths and a rest or no note. I used the lettered keypads as well as the 9 which looks like a small G for all the notes which are seven in number, basically **A B C D E F** and **G.**

Tunesmith Operation

Once you start the program, you press one of the note letters. It will sound the appropriate note. If you want the sharp for that note, if it has one (B and E do not), press **5.** To get the upper octave of the note you want you press **7,** and if you want the upper octave sharp of the note, press **5** first, then **7.** The keys **1, 2, 4,**and **8** will give you a whole note **(1),** a half note **(2),** a quarter note **(4),** and an eighth note **(8).** After you choose your note, you choose your length. If you don't want the note, start again, only this time the length is not automatically a half note as it would be when you first start out, you'll have to change it to what you want.

Now that you have your nice note that sounds just right, press **3.** This will save the note and place it in a tune table. To know that the note is indeed saved, the display will flash a **SAVE.** You have to hold the **3** key down until the **SAVE** is seen, though. Now the chosen note will be played and you can pick another note, or a rest which is **0.** The procedure is the same for a possible 72 note tune. If you like your tune and want to write it down, press the **+** key. The display will show you the first note of the tune, and every time you hit the **3** key, the next will be displayed. If you want to start again, press the **DA** (Do Again) key.

The Tunesmith Program

We can go over the program now. Table I is a listing of the keypad numbers and what they represent. The main program starts at 0200 and initialization goes on to 021A. From 021C to 0228 we test the keypad and 022A to 022E we test for the first time through the program. This step eliminates any noise in the speaker while choosing the first note. 0230 to 0236 gets the program to step through all the notes, and 0238 to 023D delays the program, not only to give you more time to choose a note, but also to put a space between the beginning and ending of the tune. 0242 to 0248 is for the beginning silence. 024A thru 0263 loads the note you have chosen into a temporary location. 0265 to 026E will jump to all the subroutines which we'll explain in a minute. 0271 thru 027B tests for the save key, which you press if you want that particular note. From 027B to 0283 we test for the **DA** key. 0285 to 028F will cause the program to jump to the routine which will allow us to see what notes we have so that they can be written down and saved for the "Top Ten". 0295 to 02A9 sets the save flag, resets the note counter, and because the program goes deep into the stack territory, resets the stack pointer to avoid trouble.

The Get High subroutine is the first one we come to. From 0356 to 035E we test to see if we want a high note. If we don't, we return from the subroutine. If yes, we'll first test to see if it's to be a sharp note that is to go to the next octave. If it is, then from 0366 to 036A we'll load the high sharp note into the temporary location, otherwise from 036F to 0373 we'll load just the next octave note. The Get Sharp subroutine is similar and the Get Length subroutine is simple enough.

The Play Tune subroutine is next. From 0300 to 0306 we set up the first note, then we play it. This is the unsaved note we are trying out. Then we'll test for a save flag from 0313 to 0317, and test for a note or notes in the tunetable up to 031D. If there is one or more notes in the tunetable, from 031F to 0330 we'll play them. If we had a save for the temporary note, we reset the save flag, store a rest so we don't hear the saved note twice, then load the note into the next position of the tuntable, and we'll also put our chosen length into the length table; all this from 0333 to 0345. Since we saved the note, not only do we need some indication that it was saved, we also need to indicate that our finger is on the **3** keypad long enough for the program to catch the keypad entry, so at 0347 we go to the subroutine that displays a big red **"SAVE".** At 034A we play all our notes again, and then go back to the main program to get another note, then back here again so we always hear our tune.

In the Tone subroutine, at 02DD and 02DF we set the ports to outputs; and at 02E2 and E4 we start KIM's internal timer. We load the note frequency, and when it runs down we change the output to its other state, whatever it was. If you hook a speaker circuit on the port as in the KIM manual, a note will be produced as we repeat this procedure every time the timer times out at 02EF; and if we do

this for a length of time determined by the note length at 02F9, we have just played a note in our tables or one we're testing out.

Our Save subroutine starts at 03AA where we load a number for a particular time we want to keep the SAVE letters on. Next at 03AE and 03BO we set the direction registers and since we want only 4 digits lit we load the number 4 into the X register. When we store one of six numbers, from 09 to 13 into the location SBD(1742), one of the six digits will be lit, and then if we load a particular hex number representing a letter, number or other shape into another location SAD(1740), then the seven segment display will light. We also need some delay, because if we did not, the display would light and go out in a couple of microseconds, which few of us could see. All this is taken care of from 03B3 to 03CC. And finally we want to end the tune after 72 notes so we will automatically go the the Display Notes routine from 03CE to 03D4. We want to keep count of how many notes we save so at 03D7 we increment the note count.

If we have a nice little tune running through our circuits and we say to ourselves, "Hey, that's a catchy tune that might make the top 40," then we'll need some way of finding out what notes are in the tunetable so that we can write them down. The Display Notes routine does just that. What we want this section to do is to display a lettered note, to show that it is a sharp and/or a high note, and to show what its length is. We want it to stay on the display until we're ready for the next note and we need some indication that the note has changed when we do go to the next note. Finally we want the option of starting again. So here we go.

From 0100 to 010A we test the counters to see if we've reached the end of our tune table, then we take our note and length and put them into a temporary location from 010D to 0115. From 0117 to 011D we check for a rest; if it isn't one then at 011F on we determine what note it is. What I did was to compare the unknown note to the note table and for every wrong comparison increment a count. We also have four groups of 7 notes and to determine what group, I subtract a number until I get a carry flag. This then tells me the group and also the note. The group indicates whether the note is high, sharp, or high/sharp. We load the correct shape for the display on this information. If it was just a rest, at 0180 we load a zero shape. At 018A to 0198 we test for the length and then store the length shape. Up to 01BC we display the shapes as before, only this time, as we go through a test for the next note, and "do again", we keep the

```
                        TUNESMITH

                        BY ANTHONY T. SCARPELLI
                        MAY 1979

                        MICRO NUMBER 13
                        JUNE 1979
                        COPYRIGHT (C) MAY 1979 BY
                        THE COMPUTERIST, INC.

                        KIM MONITOR REFERENCES

                        PAD    *    $1700  DATA REGISTER
                        PADD   *    $1701  DATA DIRECTION REGISTER
                        TIMER  *    $1704  SET TIMER
                        TTIMER *    $1707  TEST TIMER
                        STIMER *    $170F  START TIMER
                        SAD    *    $1740  SYSTEM DATA REGISTER A
                        SADD   *    $1741  SYSTEM DATA DIRECTION A REG
                        SBD    *    $1742  SYSTEM DATA REGISTER B
                        PBDD   *    $1743  SYSTEM DATA DIRECTION REG B
                        KEYIN  *    $1F40  KEYPAD INPUT
                        GETKEY *    $1F6A  GET KEYBOARD INPUT

                        PAGE ZERO LOCATIONS

0000                            ORG   $0000

                        LOW NOTE TABLE

0000 FB                 NOTE    =     $FB    G
0001 DF                         =     $DF    A
0002 C6                         =     $C6    B
0003 BB                         =     $BB    C
0004 A6                         =     $A6    D
0005 93                         =     $93    E
0006 8A                         =     $8A    F

                        HIGH NOTE TABLE

0007 7B                 HINOTE  =     $7B    G
0008 6D                         =     $6D    A
0009 61                         =     $61    B
000A 5B                         =     $5B    C
000B 51                         =     $51    D
000C 48                         =     $48    E
000D 43                         =     $43    F

                        LOW SHARP NOTE TABLE

000E ED                 SHPNOT  =     $ED    G SHARP, A FLAT
000F D2                         =     $D2    A SHARP, B FLAT
0010 01                         =     $01    NO NOTE
0011 B0                         =     $B0    C SHARP, D FLAT
0012 9C                         =     $9C    D SHARP, E FLAT
0013 01                         =     $01    NO NOTE
0014 83                         =     $83    F SHARP, G FLAT

                        HIGH SHARP NOTE TABLE

0015 74                 HISHRP  =     $74    G SHARP, A FLAT
0016 67                         =     $67    A SHARP, B FLAT
0017 01                         =     $01    NO NOTE
0018 56                         =     $56    C SHARP, D FLAT
```

```
0019 4C              =     $4C    D SHARP, E FLAT
001A 01              =     $01    NO NOTE
001B 3F              =     $3F    F SHARP, G FLAT

001C 00              =     $00    UNUSED
001D 00              =     $00
001E 00              =     $00
001F 00              =     $00

0020 02      DELTIM  =     $02    DELAY TIME
0021 00      TIMED   =     $00
0022 00      TIMEC   =     $00
0023 00      SAVFLG  =     $00    SAVE FLAG
0024 00      TLENTH  =     $00    TEMP. LENGTH
0025 00      NOTPTR  =     $00    NOTE POINTER
0026 00      KEYPTR  =     $00    KEY POINTER
0027 00      TNOTE   =     $00    TEMP NOTE
0028 00      HIFLG   =     $00    HIGH FLAG
0029 00      SHPFLG  =     $00    SHARP FLAG
002A 00      NOTNUM  =     $00    NOTE NUMBER
002B 00      PRMNOT  =     $00    PERMANENT NOTE
002C 00      FSTFLG  =     $00    FIRST TIME FLAG
002D 00      PLENTH  =     $00    PERM. LENGTH
002E 00      TNTNUM  =     $00    TEMP. NOTE NUMBER
002F 00      NEXNOT  =     $00    NEXT NOTE
0030 00      DELAYA  =     $00    DELAY A
0031 00      DELAYB  =     $00    DELAY B
0032 00      PNTPTR  =     $00    PERM. NOTE POINTER
0033 00      DELAYC  =     $00    DELAYC
0034 00      TTBPTR  =     $00    TUNETABLE POINTER
0035 00      NTBPTR  =     $00    NOTE TABLE POINTER
0036 00      NOTCNT  =     $00    NOTCNT NOTE COUNT
0037 00      DNTCNT  =     $00    DISPLAY NOTE COUNT
0038 00      TEMNOT  =     $00    TEMP. NOTE
0039 00      TEMLEN  =     $00    TEMP. LENGTH
003A 00      COUNT   =     $00
003B 00      DFOUR   =     $00
003C 00      DTHREE  =     $00
003D 00      DTWO    =     $00
003E 00      DONE    =     $00
003F 00      LNTPTR  =     $00    LENGTH POINTER

             CONSTANTS

0040 01      KEYLNT  =     $01    (1) WHOLE NOTE
0041 02              =     $02    (2) HALF NOTE
0042 04              =     $04    (4) QUARTER NOTE
0043 08              =     $08    (8) EIGHTH NOTE
0044 20      LNTH    =     $20    LENGTH
0045 10              =     $10
0046 08              =     $08
0047 04              =     $04
0048 86      LNSHP   =     $86    (1) LENGTH SHAPE
0049 DB              =     $DB    (2)
004A E6              =     $E6    (4)
004B FF              =     $FF    (8)
004C BD      NTSHP   =     $BD    (G) LETTER SHAPES
004D F7              =     $F7    (A)
004E FC              =     $FC    (B)
004F B9              =     $B9    (C)
0050 DE              =     $DE    (D)
0051 F9              =     $F9    (E)
0052 F1              =     $F1    (F)

0053 00      LETNUM  =     $00    LETTER NUMBER
```

display lit. If we hit the **3** key we jump to a delay which blanks the display. This lets us know a new note has entered the circuits so that we can distinguish two or more same notes in a row. Finally we reset the stack pointer again and display the next note. If we want to start again at any time, we hit the **DA** key and off we go to the beginning again. By the way, the delay subroutine we go to is a good delay to get very long times. It uses the KIM-1's internal timer.

So that's it. I know it is a long program, because of all the explanation, but I want as *much understanding as possible*, because of the possibilities it holds. The simple tone generation can be replaced with a D/A converter, an erase note mode can be implemented, a larger scale with more lengths and other variables can be developed, and so on. There is no limit. But for a beginning, with a small computer, all you potential Bachs, here it is, go to it.

μ

Table I — Keypad Representations

A = A note
B = B note
C = C note
D = D note
E = E note
F = F note
9 = G note
0 = rest
1 = whole note
2 = 1/2 note
4 = 1/4 note
8 = 1/8 note
5 = sharp
7 = upper octave
3 = save or display next note
DA = Do Again
+ = Display notes

Classified Ads

APPLE RENUMBER/APPEND - Integer and Applesoft! Programmer's Utility Pack. $19.95 for disk or tape. Includes many other programs as well. SASE for info or order from:
Southwestern Data Systems
Box 582-M
Santee, CA 92071
714/562-3670

Amateur Radio LOG Program for the Apple. Use for QSO's, contests, field days, DX, etc. Has 7 functions. Full info for revising to your needs. Cassette $12.00. Order from:
Alex Massimo, A F 6 W
4041 41st Street
San Diego, CA 92105

PROFIT from your micro. Don Lancaster's outrageous new book THE INCREDIBLE SECRET MONEY MACHINE tells, shows you how. $6.95 autographed, postpaid, guaranteed. Visa Accepted. Quest your tinaja NOW!
Order from:
Synergetics MC-7
Box 1877
Thatcher, AZ 85552

COMPUTERS IN PSYCHIATRY/PSYCHOLOGY (formerly MICRO-PSYCH) Bimonthly magazine: articles, program catalogue, bibliography, training and employment clearinghouse, calendar, more. Over 300 subscribers. Send $12.00 for Vol 1 (1978-1979); $15.00 for Vol 2 (1979-1980). Order from:
COMPUTERS IN PSYCHIATRY/PSYCHOLOGY
Box I
26 Trumbull Street
New Haven, CT 06511

MICRO

```
0054 0F                =      $0F
0055 0D                =      $0D
0056 0B                =      $0B
0057 09                =      $09

0058 00         LETTER =      $00     LETTER SHAPES
0059 F9                =      $F9
005A BE                =      $BE
005B F7                =      $F7
005C ED                =      $ED

005D            TUNTBL *      $0060   TUNE TABLE
005D            LNTTBL *      $00A8   LENGTH TABLE



                DISPLAY NOTE ROUTINE

0100                   ORG    $0100

0100 A9 01      DISNTS LDAIM  $01     RESET DISPLAY NOTE COUNT
0102 85 37             STA    DNTCNT
0104 A5 37      NXTNOT LDA    DNTCNT  TEST FOR END
0106 C5 36             CMP    NOTCNT
0108 D0 03             BNE    BEGIN
010A 4C E1 01          JMP    DOAGNB
010D A6 37      BEGIN  LDX    DNTCNT  STORE NOTE
010F B5 60             LDAZX  TUNTBL  AND LENGTH
0111 85 38             STA    TEMNOT
0113 B5 A8             LDAZX  LNTTBL
0115 85 39             STA    TEMLEN
0117 A2 00             LDXIM  $00
0119 A5 38      RPT    LDA    TEMNOT  TEST FOR TEST
011B C9 01             CMPIM  $01
011D F0 61             BEQ    DISZER
011F D5 00             CMPZX  NOTE    TEST FOR NOTE
0121 F0 04             BEQ    SUB
0123 E8                INX
0124 4C 19 01          JMP    RPT
0127 38         SUB    SEC            TEST FOR FIRST
0128 8A                TXA            GROUP
0129 E9 07             SBCIM  $07
012B B0 0D             BCS    NXGRPA
012D B5 4C             LDAZX  NTSHP   STORE NOTE SHAPE
012F 85 3D             STA    DTWO
0131 A9 C0             LDAIM  $C0
0133 85 3E             STA    DONE
0135 85 3C             STA    DTHREE
0137 4C 8A 01          JMP    DISLEN

013A 38         NXGRPA SEC            TEST FOR SECOND
013B 8A                TXA            GROUP
013C E9 0E             SBCIM  $0E
013E B0 13             BCS    NXGRPB
0140 8A                TXA
0141 E9 06             SBCIM  $06     STORE NOTE SHAPE
0143 AA                TAX
0144 B5 4C             LDAZX  NTSHP
0146 85 3D             STA    DTWO
0148 A9 F6             LDAIM  $F6     STORE HI SHAPE
014A 85 3E             STA    DONE
014C A9 C0             LDAIM  $C0
014E 85 3C             STA    DTHREE
0150 4C 8A 01          JMP    DISLEN
```

```
0153 38        NXGRPB SEC          TEST FOR THIRD
0154 8A               TXA          GROUP
0155 E9 15            SBCIM $15
0157 B0 13            BCS   NXGRPC
0159 8A               TXA
015A E9 0D            SBCIM $0D    STORE NOTE SHAPE
015C AA               TAX
015D B5 4C            LDAZX NTSHP
015F 85 3D            STA   DTWO
0161 A9 ED            LDAIM $ED
0163 85 3C            STA   DTHREE
0165 A9 C0            LDAIM $C0
0167 85 3E            STA   DONE
0169 4C 8A 01         JMP   DISLEN

016C 38        NXGRPC SEC          STORE NOTE SHAPE
016D 8A               TXA
016E E9 15            SBCIM $15
0170 AA               TAX
0171 B5 4C            LDAZX NTSHP
0173 85 3D            STA   DTWO
0175 A9 ED            LDAIM $ED    STORE SHARP SHAPE
0177 85 3C            STA   DTHREE
0179 A9 F6            LDAIM $F6
017B 85 3E            STA   DONE
017D 4C 8A 01         JMP   DISLEN

0180 A9 BF     DISZER LDAIM $BF    STORE ZERO SHAPE
0182 85 3D            STA   DTWO
0184 A9 C0            LDAIM $C0
0186 85 3E            STA   DONE
0188 85 3C            STA   DTHREE
018A A2 00     DISLEN LDXIM $00
018C A5 39     RPTB   LDA   TEMLEN TEST FOR LENGTH
018E D5 44            CMPZX LNTH
0190 F0 04            BEQ   GTSHP
0192 E8               INX
0193 4C 8C 01         JMP   RPTB
0196 B5 48     GTSHP  LDAZX LNSHP  STORE LENGTH SHAPE
0198 85 3B            STA   DFOUR
019A A9 80     DIS    LDAIM $80    LOAD DISPLAY
019C 85 33            STA   DELAYC LIGHT TIME
019E A9 7F            LDAIM $7F    SET DIRECTION REGISTER
01A0 8D 41 17         STA   SADD
01A3 A2 04     RPTC   LDXIM $04    SET UP 4 LETTERS
01A5 A0 FF     LITE   LDYIM $FF    AND DISPLAY
01A7 B5 53            LDAZX LETNUM LIGHT LETTERS
01A9 8D 42 17         STA   SBD
01AC B5 3A            LDAZX COUNT
01AE 8D 40 17         STA   SAD
01B1 88        WAIT   DEY          DELAY
01B2 D0 FD            BNE   WAIT
01B4 CA               DEX          GET NEXT LETTER
01B5 10 EE            BPL   LITE
01B7 A4 33            LDYZ  DELAYC DELAY
01B9 88               DEY
01BA 84 33            STYZ  DELAYC
01BC D0 E5            BNE   RPTC
01BE 20 40 1F         JSR   KEYIN  TEST FOR NEXT NOTE
01C1 20 6A 1F         JSR   GETKEY
01C4 C9 03            CMPIM $03
01C6 F0 0C            BEQ   NEXT
01C8 20 40 1F         JSR   KEYIN  TEST FOR START AGAIN
01CB 20 6A 1F         JSR   GETKEY
01CE C9 11            CMPIM $11
```

Classified Ads

FREE SAMPLE COPY - New BUY-SELL-TRADE Publication, indexed by product type, mailed monthly First Class. 3500 circ., low, low ad rates. Send for your free sample copy now! Subscription $3.50 per year, FREE THREE LINE AD with subscription. Order from:
LECTRONICS EMPORIUM
Dick Costello, Pub.
P.O. Box 828
Derry, NH 03038

The TARGET for users of Rockwell's AIM 65. Find out how to use the printer, keyboard and display. Reviews of upcoming Assembler and BASIC in ROMs. Six bimonthly issues for $5.00 US and Canada ($12.00 elsewhere). Order from:
Don Clem
RR#2
Spencerville, OH 45887

ZIPTAPE loads 8K BASIC in 15 seconds! Slower than a speeding disc? Sure, but it only costs $22.50 plus $1.00 S&H. $3.00 extra for software on KIM cassette Described in MICRO #6. SASE for info. Order from:
Lew Edwards
1451 Hamilton Ave.
Trenton, NJ 08629

New! APPLE SOFTWARE DIRECTORY has over 700 programs and sources listed alphabetically. Order Vol 1 - Business, Vol 2 - Games. Dealer program available. Each book $4.95 at Apple dealers or order from:
WIDL Video
5325 N. Lincoln
Chicago, IL 60625

MICRO

```
01D0 F0 0F           BEQ   DOAGNB
01D2 D0 C6           BNE   DIS

01D4 20 AC 02  NEXT  JSR   DELAY
01D7 E6 37           INC   DNTCNT  INCREMENT DISPLAY NOTE
01D9 A2 FF           LDXIM $FF     COUNT.  RESET STACK
01DB 9A              TXS           POINTER
01DC EA              NOP           PADDING
01DD EA              NOP
01DE 4C 04 01        JMP   NXTNOT

01E1 A9 00     DOAGNB LDAIM $00
01E3 85 36           STA   NOTCNT
01E5 4C 00 02        JMP   NUTUNE


15 16 17
               MAIN PROGRAM

0200                 ORG   $0200

0200 A9 00     NUTUNE LDAIM $00     INITIALIZE TUNE
0202 85 23           STA   SAVFLG
0204 85 2A           STA   NOTNUM
0206 85 2C           STA   FSTFLG
0208 A9 01           LDAIM $01
020A 85 60           STA   TUNTBL
020C 85 A8           STA   LNTTBL
020E 85 27           STA   TNOTE
0210 A9 10           LDAIM $10
0212 85 24           STA   TLENTH
0214 A9 06     NUNOTE LDAIM $06     INITIALIZE NOTE
0216 85 25           STA   NOTPTR
0218 A9 0F           LDAIM $0F
021A 85 26           STA   KEYPTR
021C 20 40 1F  PLAYB JSR   KEYIN   TEST KEYPAD FOR NOTE
021F 20 6A 1F        JSR   GETKEY
0222 C5 26           CMP   KEYPTR
0224 F0 2D           BEQ   GTNOTE
0226 C9 00           CMPIM $00     FOR REST
0228 F0 20           BEQ   GTREST
022A A5 2C           LDA   FSTFLG  TEST FOR FIRST TIME
022C C9 00           CMPIM $00
022E F0 12           BEQ   NOPLAY
0230 C6 26           DEC   KEYPTR  SET UP FOR NEXT NOTE
0232 C6 25           DEC   NOTPTR
0234 10 02           BPL   DELYA
0236 30 DC           BMI   NUNOTE

0238 A6 30     DELYA LDXZ  DELAYA DELAY
023A CA              DEX
023B 86 30           STXZ  DELAYA
023D D0 DD           BNE   PLAYB
023F 4C 65 02        JMP   SVNOTE

0242 C6 26     NOPLAY DEC  KEYPTR SET UP FOR NEXT NOTE
0244 C6 25           DEC   NOTPTR
0246 10 D4           BPL   PLAYB
0248 30 CA           BMI   NUNOTE

024A A9 01     GTREST LDAIM $01     LOAD REST
024C 85 2C           STA   FSTFLG
024E 85 27           STA   TNOTE
0250 4C 65 02        JMP   SVNOTE
```

```
0253 A9 01     GTNOTE LDAIM $01    LOAD FIRST NOTE FLAG
0255 85 2C            STA   FSTFLG
0257 A6 25            LDXZ  NOTPTR LOAD CHOSEN NOTE
0259 A9 00            LDAIM $00
025B 85 28            STA   HIFLG
025D 85 29            STA   SHPFLG
025F B5 00            LDAZX NOTE
0261 85 27            STA   TNOTE
0263 86 32            STXZ  PNTPTR
0265 20 56 03  SVNOTE JSR   GETHI  GET HIGH NOTE
0268 20 86 03         JSR   GETSRP GET SHARP NOTE
026B 20 DA 03         JSR   GTLNTH GET LENGTH
026E 20 00 03         JSR   PLATUN PLAY NOTE
0271 20 40 1F         JSR   KEYIN  TEST TO SAVE NOTE
0274 20 6A 1F         JSR   GETKEY
0277 C9 03            CMPIM $03
0279 F0 16            BEQ   SAVE
027B 20 40 1F         JSR   KEYIN  TEST OFR START OVER
027E 20 6A 1F         JSR   GETKEY
0281 C9 11            CMPIM $11    DA = DO AGAIN
0283 F0 13            BEQ   DOAGN
0285 20 40 1F         JSR   KEYIN  TEST FOR DIPSLAY NOTER
0288 20 6A 1F         JSR   GETKEY
028B C9 12            CMPIM $12    = +
028D F0 15            BEQ   DNOTES
028F D0 8B            BNE   PLAYB

0291 A9 01     SAVE   LDAIM $01    SAVE NOTE
0293 85 23            STA   SAVFLG
0295 4C 14 02         JMP   NUNOTE

0298 A9 00     DOAGN  LDAIM $00    RESET NOTE COUNTER
029A 85 36            STA   NOTCNT
029C A2 FF            LDXIM $FF    RESET STACK POINTER
029E 9A               TXS
029F EA               NOP          PADDING
02A0 EA               NOP
02A1 4C 00 02         JMP   NUTUNE

02A4 A2 FF     DNOTES LDXIM $FF    RESET STACK POINTER
02A6 9A               TXS
02A7 EA               NOP
02A8 EA               NOP
02A9 4C 00 01         JMP   DISNTS JUMP TO DISPLAY NOTES

               DELAY SUBROUTINE

02AC A5 20     DELAY  LDA   DELTIM GET DELAY VALUE
02AE 85 21            STA   TIMED
02B0 A9 FF     DELA   LDAIM $FF    LOAD TIMER
02B2 8D 04 17         STA   TIMER
02B5 2C 07 17  TEST   BIT   TTIMER TEST TIMER
02B8 10 FB            BPL   TEST   BRANCH IF NOT RUN OUT
02BA C6 22            DEC   TIMEC  REDUCE TIME VALUE
02BC D0 F2            BNE   DELA   START AGAIN
02BE C6 21            DEC   TIMED  REDUCE DELAY VALUE
02C0 D0 EE            BNE   DELA   BRANCH IF NOT DNOE
02C2 60               RTS



02DD                  ORG   $02DD

               TONE SUBROUTINE
```

```
02DD A9 01      TONE    LDAIM $01     OPEN PORT
02DF 8D 01 17           STA   PADD
02E2 A9 20      SOUND   LDAIM $20     START TIMER
02E4 8D 0F 17           STA   STIMER
02E7 A6 2B      NOTEX   LDXZ  PRMNOT  NOTE FREQUENCY
02E9 CA         NWAIT   DEX
02EA D0 FD              BNE   NWAIT
02EC EE 00 17           INC   PAD     TOGGLE OUTPUT
02EF A9 80              LDAIM $80     TEST COUNTER
02F1 2C 07 17           BIT   TTIMER
02F4 30 03              BMI   TIMOUT
02F6 4C E7 02           JMP   NOTEX
02F9 C6 2D      TIMOUT  DEC   PLENTH  NOTE LENGTH
02FB D0 E5              BNE   SOUND
02FD 60                 RTS


                        PLAY TUNE SOBROUTINE

0300                    ORG   $0300

0300 A5 2A      PLATUN  LDA   NOTNUM  SET UP FIRST NOTE
0302 85 2E              STA   TNTNUM
0304 A9 00              LDAIM $00
0306 85 2F              STA   NEXNOT
0308 A5 27              LDA   TNOTE   PLAY NOTE
030A 85 2B              STA   PRMNOT
030C A5 24              LDA   TLENTH
030E 85 2D              STA   PLENTH
0310 20 DD 02           JSR   TONE
0313 A5 23              LDA   SAVFLG  TEST FOR SAVE
0315 C9 01              CMPIM $01
0317 F0 1A              BEQ   SAVEX
0319 A5 2A              LDA   NOTNUM  TEST FOR NOTE
031B C9 00              CMPIM $00     (NOT REQUIRED)
031D F0 13              BEQ   RETURN
031F A6 2F      PLAYC   LDXZ  NEXNOT  LOAD NEXT NOTE
0321 B5 60              LDAZX TUNTBL
0323 85 2B              STA   PRMNOT
0325 B5 A8              LDAZX LNTTBL  LOAD NEXT LENGTH
0327 85 2D              STA   PLENTH
0329 20 DD 02           JSR   TONE    PLAY NOTE
032C E6 2F              INC   NEXNOT  SET UP FOR
032E C6 2E              DEC   TNTNUM  NEXT NOTE
0330 10 ED              BPL   PLAYC
0332 60         RETURN  RTS

0333 A9 00      SAVEX   LDAIM $00     RESET SAVE FLAG
0335 85 23              STA   SAVFLG
0337 A9 01              LDAIM $01     NO PLAY
0339 85 27              STA   TNOTE
033B E6 2A              INC   NOTNUM  LOAD NOTE INTO
033D A6 2A              LDXZ  NOTNUM  TUNETABLE
033F A5 2B              LDA   PRMNOT
0341 95 60              STAZX TUNTBL
0343 A5 24              LDA   TLENTH  LOAD LENGTH
0345 95 A8              STAZX LNTTBL  INTO LENGTH TABLE
0347 20 AA 03           JSR   DISPLY
034A 4C 00 03           JMP   PLATUN


0356                    ORG   $0356
```

```
                    GET HIGH SUBROUTINE

0356 20 40 1F  GETHI  JSR   KEYIN  TEST FOR HIGH NOTE
0359 20 6A 1F         JSR   GETKEY
035C C9 07            CMPIM $07
035E D0 15            BNE   RETRNB
0360 A5 29            LDA   SHPFLG TEST SHARP NOTE
0362 C9 00            CMPIM $00    (NOT REQUIRED)
0364 F0 09            BEQ   LOADHI
0366 A6 32            LDXZ  PNTPTR LOAD HIGH SHARP NOTE
0368 B5 15            LDAZX HISHRP
036A 85 27            STA   TNOTE
036C 4C 75 03         JMP   RETRNB (COULD HAVE BEEN RTS)
036F A6 32     LOADHI LDX   PNTPTR LOAD HIGH NOTE
0371 B5 07            LDAZX HINOTE
0373 85 27            STA   TNOTE
0375 60        RETRNB RTS



0386                  ORG   $0386

                    GET SHARP SUBROUTINE

0386 20 40 1F  GETSRP JSR   KEYIN  TEST FOR SHARP NOTE
0389 20 6A 1F         JSR   GETKEY
038C C9 05            CMPIM $05
038E D0 0A            BNE   RETRNC
0390 A9 01            LDAIM $01    LOAD SHARP FLAG
0392 85 29            STA   SHPFLG
0394 A6 32            LDXZ  PNTPTR LOAD SHARP NOTE
0396 B5 0E            LDAZX SHPNOT
0398 85 27            STA   TNOTE
039A 60        RETRNC RTS



03AA                  ORG   $03AA

                    DISPLAY SAVE SUBROUTINE

03AA A9 80     DISPLY LDAIM $80    LOAD DISPLAY
03AC 85 33            STA   DELAYC LIGHT TIME
03AE A9 7F            LDAIM $7F    SET DIRECTION REGISTER
03B0 8D 41 17         STA   SADD
03B3 A2 04     REPEAT LDXIM $04    SET UP 4 LETTERS
03B5 A0 FF     LIGHT  LDYIM $FF    AND DELAY
03B7 B5 53            LDAZX LETNUM LIGHT LETTERS
03B9 8D 42 17         STA   SBD
03BC B5 58            LDAZX LETTER
03BE 8D 40 17         STA   SAD
03C1 88        WAITY  DEY          DELAY
03C2 D0 FD            BNE   WAITY
03C4 CA               DEX          GET NEXT LETTER
03C5 10 EE            BPL   LIGHT
03C7 A4 33            LDY   DELAYC DELAY
03C9 88               DEY
03CA 84 33            STY   DELAYC
03CC D0 E5            BNE   REPEAT
03CE A5 36            LDA   NOTCNT TEST FOR 72 NOTES
03D0 C9 48            CMPIM $48    48 HEX = 72 DECIMAL
03D2 D0 03            BNE   INCNOT
03D4 4C 00 01         JMP   DISNTS
```

MICROBES

LIFESAVER - MICRO 11

11:10 There should be vertical characters in lines 230 to 290 similar to those in line 220. Also, each line needs a " at the end. Line 310 should read "homeGOTO" not "home GOTO". The correct pattern for the Cheshire Cat is:

AN APPLE II PROGRAM EDIT AID

11:5 Line 32630 should read J=ASC ... not I=ASC ...

"And here is an addition to make the program run smoother: Add the following lines so that when the screen fills up with text, the Apple will pause and display an "@" in the lower right-hand corner. This will prompt you to hit any key and Apple will clear the screen and continue wherehit left off. This process will continue until until there are no more occurences of the search item.

Change: 32690
IF PEEK(37)<18 THEN 32700

Add:
32692 VTAB 23:TAB 39:PRINT"@"
32695 KEY=PEEK(-16384): IF KEY <127 THEN 32695
32697 POKE -16368,0: CALL -936

David B. Garson
5163 Willow Wood Road
Rolling Hills Estates, CA 90274

The Ultimate PET Renumber

11:37 A few listing errors:

1F15 88 DEY was C8
1F49 A5 0A LDAZ was A9 0A
1F4B F0 2F BEQ INSF was F0 17

```
03D7 E6 36      INCNOT INC   NOTCNT INCREASE NOTE COUNT
03D9 60                RTS

                       GET LENGTH SUBROUTINE

03DA A9 03      GTLNTH LDAIM $03    LOAD LENGTH POINTER
03DC 85 3F             STA   LNTPTR
03DE 20 40 1F   KEYTST JSR   KEYIN  TEST KEYPAD FOR
03E1 20 6A 1F          JSR   GETKEY LENGTH
03E4 A6 3F             LDXZ  LNTPTR
03E6 D5 40             CMPZX KEYLNT
03E8 F0 05             BEQ   LODLNT
03EA C6 3F             DEC   LNTPTR
03EC 10 F0             BPL   KEYTST
03EE 60                RTS
03EF B5 44      LODLNT LDAZX LNTH   LOAD LENGTH
03F1 85 24             STA   TLENTH
03F3 60                RTS
```

```
SYMBOL TABLE 2000 228E
NOTE   0000   HINOTE 0007   SHPNOT 000E   HISHRP 0015
DELTIM 0020   TIMED  0021   TIMEC  0022   SAVFLG 0023
TLENTH 0024   NOTPTR 0025   KEYPTR 0026   TNOTE  0027
HIFLG  0028   SHPFLG 0029   NOTNUM 002A   PRMNOT 002B
FSTFLG 002C   PLENTH 002D   TNTNUM 002E   NEXNOT 002F
DELAYA 0030   DELAYB 0031   PNTPTR 0032   DELAYC 0033
TTBPTR 0034   NTBPTR 0035   NOTCNT 0036   DNTCNT 0037
TEMNOT 0038   TEMLEN 0039   COUNT  003A   DFOUR  003B
DTHREE 003C   DTWO   003D   DONE   003E   LNTPTR 003F
KEYLNT 0040   LNTH   0044   LNSHP  0048   NTSHP  004C
LETNUM 0053   LETTER 0058   TUNTBL 0060   LNTTBL 00A8
DISNTS 0100   NXTNOT 0104   BEGIN  010D   RPT    0119
SUB    0127   NXGRPA 013A   NXGRPB 0153   NXGRPC 016C
DISZER 0180   DISLEN 018A   RPTB   018C   GTSHP  0196
DIS    019A   RPTC   01A3   LITE   01A5   WAIT   01B1
NEXT   01D4   DOAGNB 01E1   NUTUNE 0200   NUNOTE 0214
PLAYB  021C   DELYA  0238   NOPLAY 0242   GTREST 024A
GTNOTE 0253   SVNOTE 0265   SAVE   0291   DOAGN  0298
DNOTES 02A4   DELAY  02AC   DELA   02B0   TEST   02B5
TONE   02DD   SOUND  02E2   NOTEX  02E7   NWAIT  02E9
TIMOUT 02F9   PLATUN 0300   PLAYC  031F   RETURN 0332
SAVEX  0333   GETHI  0356   LOADHI 036F   RETRNB 0375
GETSRP 0386   RETRNC 039A   DISPLY 03AA   REPEAT 03B3
LIGHT  03B5   WAITY  03C1   INCNOT 03D7   GTLNTH 03DA
KEYTST 03DE   LODLNT 03EF   PAD    1700   PADD   1701
TIMER  1704   TTIMER 1707   STIMER 170F   SAD    1740
SADD   1741   SBD    1742   PBDD   1743   KEYIN  1F40
GETKEY 1F6A
```